

Полезные, но редко используемые, возможности команды *plot*

Система Maple умеет работать с величинами, которые имеют размерность. Эта размерность может выводиться на графиках, но в том виде, как принято в международных стандартах.


Пусть *dist* – расстояние в километрах, которое прошёл пешеход за время *t*.

```
> dist := (6*t^2+5*t) / (t+2);
```

$$dist := \frac{6t^2 + 5t}{t + 2}$$

Соответствующая размерность может быть отображена вместо одной из подписей по координатной оси или же вместе с этой подписью.


```
> plot(dist*Unit('km'), t=0..3);
```



В этом примере явно указано, что аргумент функции задан в радианах.


Координатные оси смещены к краям рисунка.

```
> plot(sin(x)^3+cos(x)^3, x=-Pi..Pi*Unit('radians'), axes=frame);
```



Если есть риск, что изображение будет сильно сжато, то лучше увеличить размер шрифта, которым отображаются масштабные метки посредством опции *axesfont*. В старых версиях Maple это приводило к неожиданным эффектам.


```
> plot(sin(x)^3+cos(x)^3,x=-Pi..Pi*Unit('radians'),  
       axes=frame,axesfont=["Helvetica",12]);
```



На количество отображаемых масштабных меток можно повлиять с помощью опции *tickmarks*, значением которой является список из двух элементов. В простейшем случае оба элемента списка являются натуральными числами. Эти числа представляют собой желаемое количество меток по каждой из координатной осей. Если Maple сочтёт эти значения неприемлемыми, то они будут заменены на ближайшие допустимые значения.


В данном случае настройка по оси абсцисс была проигнорирована.

```
> plot(sin(x)^3+cos(x)^3,x=-Pi..Pi*Unit('radians'),  
       tickmarks=[5,3],axesfont=["Helvetica",12]);
```



В нижней части рисунка можно разместить подпись к нему. Для этого предназначена опция *caption*. Чаще всего это просто строка текста.


```
> plot(sin(x)^3+cos(x)^3,x=-Pi..Pi,axesfont=["Helvetica",12],  
       caption="Сумма кубов");
```



Сумма кубов

Шрифт по умолчанию может показаться мелким. Тогда его следует изменить, применив опцию *captionfont*. Значением этой опции, как и всех других опций заканчивающихся на *font*, является список. Элементы этого списка имеют традиционный для системы Maple смысл: *гарнитура*, *начертание*, *размер*.


```
> plot(sin(x)^3+cos(x)^3, x=-Pi..Pi, axesfont=["Helvetica", 12],
      caption="Сумма кубов двух функций",
      captionfont=["Helvetica", "Bold", 12]);
```



Сумма кубов двух функций

В подпись можно добавлять математические выражения, записывая их по правилам языка системы Maple. Как уже отмечалось ранее, для этого следует воспользоваться процедурой **typeset**.

```
> plot(sin(x)^3+cos(x)^3, x=-Pi..Pi, axesfont=["Helvetica", 12],
      caption=typeset("Сумма кубов ", sin(x)^3+cos(x)^3),
      captionfont=["Helvetica", "Bold", 12]);
```



Сумма кубов $\sin(x)^3 + \cos(x)^3$


При печати расстояние между нижней частью рисунка и подписью оказывается слишком большим. Именно по этой причине данная опция употребляется достаточно редко.

По умолчанию, цвет, которым рисуется график, задаётся настройками системы. Эти настройки зависят от версии Maple. Начиная с ранних версий Maple утверждалось, что *the following color names are predefined in Maple:*

Набор стандартных цветов


<i>aquamarine</i>	<i>black</i>	<i>blue</i>	<i>navy</i>	<i>coral</i>
<i>cyan</i>	<i>brown</i>	<i>gold</i>	<i>green</i>	<i>gray</i>
<i>grey</i>	<i>khaki</i>	<i>magenta</i>	<i>maroon</i>	<i>orange</i>
<i>pink</i>	<i>plum</i>	<i>red</i>	<i>sienna</i>	<i>tan</i>
<i>turquoise</i>	<i>violet</i>	<i>wheat</i>	<i>white</i>	<i>yellow</i>

На экране данные цвета будут выглядеть следующим образом:



Как и в ранних версиях Maple данные цвета можно использовать в опции *color* без двойных кавычек.

```
> plot(sin(x)^3+cos(x)^3,x=-Pi..Pi,color=black,
caption=typeset("Сумма кубов ",sin(x)^3+cos(x)^3));
```



Сумма кубов $\sin(x)^3 + \cos(x)^3$

Исторически у серого цвета сложилось два названия: *gray* и *grey*. Можно использовать любое из них.

Современные версии Maple поддерживают несколько цветовых пространств и по возможностям манипулирования цветом не уступают графическим редакторам. С дополнительными деталями можно познакомиться в описании пакета **ColorTools**, который впервые появился в Maple 16.


Значением описанной ранее опции *filled* может быть не только логическая константа *true* или *false*. Как и в опциях, заканчивающихся на *font*, после знака равенства допускается список. Элементом этого списка вполне может быть опция *color*.

Вот как можно преобразовать пример с опцией *filled*:

```
> F:=exp(-x^2)+2*exp(-(x-3)^2);
```


$$F := e^{-x^2} + 2e^{-(x-3)^2}$$

```
> plot(F,x=-2..5,color=blue,filled=[color=cyan]);
```




Если графику требуются какие-либо пояснения, то в системе Maple для этой цели существует опция *legend*. Как и подпись к рисунку пояснение может быть обычной строкой текста.

```
> plot(F,x=-2..5,color=blue,legend="Сумма двух гауссовых кривых");
```




Однако, как и в других текстовых опциях, допускается использование математических выражений, если их помещать внутрь процедуры `typeset`.

```
> plot(F,x=-2..5,color=blue,legend=typeset("Сумма ",F));
```




Шрифт, которым выводятся пояснения, и их местоположение можно изменять при помощи опции `legendstyle`. Значением этой опции является список, который может содержать несколько элементов. Если левая часть одного из элементов списка называется *location*, то в правой части можно указать одно из четырёх значений: *top*, *bottom*, *right* и *left*.

```
> plot(F,x=-2..5,color=blue,legend=typeset("Сумма ",F),
      legendstyle=[location=top]);
```



Когда в левой части одного из элементов списка содержится слово *font*, то можно указать параметры шрифта как же, как и для всех упомянутых ранее опций, связанных со шрифтами.

```
> plot(F, x=-2..5, color=blue, legend=typeset("Сумма ", F),
      legendstyle=[location=top, font=["Times", "Roman", 12]]);
```



Пояснения к графикам и подписи к рисункам хорошо смотрятся на экране и нормально экспортируются во внешние файлы.

Однако попытка распечатать исходный документ или же сохранить его в формате PDF в версиях Maple пятилетней давности приводит к рисункам безобразного качества. (Именно такая версия и использовалась для создания этого файла.)

Таким образом, для качественной визуализации данных следует иметь достаточно новую версию системы. В версии Maple 2016 подобных нелепостей уже **нет**.